NASA TECHNICAL MEMORANDUM

NASA TM X-1502

NASA TM X-1502

[redacted]

COMPUTER PROGRAM FOR DESIGN OF TWO-DIMENSIONAL SUPERSONIC NOZZLE WITH SHARP-EDGED THROAT

by Michael R. Vanco and Louis J. Goldman

Lewis Research Center
Cleveland, Ohio

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • JANUARY 1968

NASA TM X-1502

COMPUTER PROGRAM FOR DESIGN OF TWO-DIMENSIONAL SUPERSONIC NOZZLE WITH SHARP-EDGED THROAT

By Michael R. Vanco and Louis J. Goldman

Lewis Research Center
Cleveland, Ohio

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

For sale by the Clearinghouse for Federal Scientific and Technical Information
Springfield, Virginia 22151 – CFSTI price $3.00

COMPUTER PROGRAM FOR DESIGN OF TWO-DIMENSIONAL SUPERSONIC NOZZLE WITH SHARP-EDGED THROAT

by Michael R. Vanco and Louis J. Goldman

Lewis Research Center

SUMMARY

The FORTRAN IV computer program for the design of a two-dimensional supersonic nozzle with a sharp-edged throat is presented along with the equations used. The nozzle, which has uniform parallel flow at the exit, is designed on the basis of two-dimensional isentropic flow of a perfect gas. This program requires as input the exit Mach number and the specific-heat ratio. The output yields the nozzle contour for the supersonic portion. Input and output for a sample case are included.

INTRODUCTION

Turbine-driven, hydrogen-fueled, open-cycle auxiliary space power systems have recently become of interest. The analysis of such systems in reference 1 indicates the possibility of using a supersonic turbine. Proper design methods must be available to obtain the highest possible efficiency from a turbine of this type.

The method of characteristics as applied to the two-dimensional isentropic flow of a perfect gas is used for the design of both supersonic nozzles and supersonic rotor blading. Such a computerized method for designing supersonic rotor blading for any Mach number level and specific-heat ratio is given in reference 2. A method for the design of sharp-edged-throat supersonic nozzles that produce uniform parallel flow at the nozzle exit is described in reference 3. This type of nozzle, an example of which is shown in reference 4, is of minimum length, which is desirable for a compact turbine. References 5 and 6 present tabulated coordinates for several sharp-edged nozzle designs for a specific-heat ratio of 1.4.

Because of the interest in supersonic turbines for auxiliary space power systems, a computer program based on the method of reference 3 was written for the general design of sharp-edged supersonic nozzles. The method of characteristics is used in the design, and the program is applicable to any Mach number level and specific-heat ratio. This report presents the equations used and the FORTRAN IV computer program along with a description of the input and output. Sample input and output are included.

SYMBOLS

A	area
A*	throat area (M = 1)
k	variable index for characteristics of family II
M	Mach number
M*	dimensionless velocity, ratio of velocity to sonic velocity at throat
m_I	slope for characteristics of family I
m_{II}	slope for characteristics of family II
n	variable index for characteristics of family I
x	x-coordinate
y	y-coordinate
γ	specific-heat ratio
μ	Mach angle, deg
ν	Prandtl-Meyer angle, angle through which flow must turn from Mach 1 to required Mach number, deg
$\Delta\nu$	increment in Prandtl-Meyer angle, deg
φ	flow angle, angle between velocity direction and x-axis, deg
I	characteristic of first family
II	characteristic of second family

Subscripts:

e	exit
k	variable index for characteristics of family II
max	maximum
n	variable index for characteristics of family I

METHOD OF ANALYSIS

The design of a two-dimensional supersonic nozzle with a sharp-edged throat described herein was based on establishing parallel uniform flow at the nozzle exit. The method of characteristics as applied to the two-dimensional isentropic flow of a perfect gas was used for the design in a manner analogous to that described in references 3 and 7. A sketch of a typical nozzle designed in this manner is shown in figure 1.



Figure 1. - Sharp-edged-throat supersonic nozzle.

Since the nozzle is symmetrical about the centerline, only half the nozzle need be designed, as shown in figure 2. The sharp-edged throat initiates a Prandtl-Meyer expan-



Figure 2. - Nomenclature and wave diagram for supersonic nozzle with sharp-edged throat.

sion (flow around a corner). The waves or characteristic lines emanating from the sharp edge have a negative slope and are termed waves of family II to be consistent with reference 7. In the region near the throat, only family II waves exist. These family II waves are then reflected at the centerline into waves with positive slopes. These waves are termed waves of family I. In the center region of the nozzle, the waves intersect, and both families of waves are present. The family I waves extend beyond the region where both families exist and intersect the nozzle contour, which is shaped so as to cancel these waves. In the exit region of the nozzle, the flow is then parallel and uniform.

The two physical boundaries for the design procedure are the nozzle contour and the centerline. The internal boundaries are the Mach lines or characteristic lines (ref. 7). Therefore, each region is bounded by either Mach lines alone, or Mach lines and a physical boundary. The equations and calculation procedures needed to design each region of the nozzle are given herein.

Equations

The nomenclature for the nozzle is given in figure 2. The supersonic portion of the nozzle is divided into a finite number of small regions in which the flow properties are considered to be constant. Each small region is denoted by two indexing variables k and n, where k is a variable index for family II characteristics and n is a variable index for family I characteristics. For nozzles with a sharp-edged throat, the contour angle at the throat is equal to one-half the exit Prandtl-Meyer angle (ref. 7), where the Prandtl-Meyer angle ν is defined as the angle through which the flow must turn from Mach 1 to the required Mach number M. The relation between ν and M is given by (ref. 5)

$$\nu = \sqrt{\frac{\gamma + 1}{\gamma - 1}} \arctan \sqrt{\left(\frac{\gamma - 1}{\gamma + 1}\right)\left(M^2 - 1\right)} - \arctan \sqrt{M^2 - 1} \tag{1}$$

The maximum values for k and n are determined by the exit Prandtl-Meyer angle and the increment in Prandtl-Meyer angle $\Delta\nu$. Therefore, the maximum values of k and n are

$$k_{max} = \frac{1}{2} \frac{\nu_e}{\Delta\nu} + 1 \tag{2}$$

and

$$n_{max} = \frac{1}{2}\frac{\nu_e}{\Delta\nu} + 1 \tag{3}$$

The flow angle φ_k for each small region is

$$\varphi_k = (k - 1)\Delta\nu \tag{4}$$

where k varies from 1 to k_{max}. The flow angle is zero near the centerline (k = 1) and $\nu_e/2$ at the wall ($k = k_{max}$).

The Prandtl-Meyer angle for each small region $\nu_{k,n}$ is

$$\nu_{k,n} = 2\Delta\nu(n - 1) + (k - 1)\Delta\nu \tag{5}$$

where n varies from 1 to n_{max}. The Prandtl-Meyer angle is zero at the throat (k = 1, n = 1) and ν_e at the nozzle exit (k = 1, $n = n_{max}$). Many of the small regions have the same Prandtl-Meyer angle, such that $\nu_{k,n} = \nu_{k-2,n+1}$. Therefore, the Prandtl-Meyer angle used in the program can be designated as ν_I, where $I = k + 2(n - 1)$.

The Mach angle for these increments is

$$\mu_{k,n} = \arcsin\left(\frac{1}{M_{k,n}}\right) \tag{6}$$

Since $\nu_{k,n}$ is known, an iterative procedure is required to solve for $M_{k,n}$. The relation between ν and $M^*_{k,n}$, the ratio of velocity to the sonic velocity at the throat, can be used. This relation is (ref. 2)

$$\nu_{k,n} = \frac{\pi}{4}\left(\sqrt{\frac{\gamma+1}{\gamma-1}} - 1\right) + \frac{1}{2}\left\{\sqrt{\frac{\gamma+1}{\gamma-1}}\arcsin\left[(\gamma - 1)M^{*2}_{k,n} - \gamma\right] + \arcsin\left(\frac{\gamma+1}{M^{*2}_{k,n}} - \gamma\right)\right\} \tag{7}$$

The bounds on $M^*_{k,n}$ of equation (7) are 1 and $\sqrt{(\gamma + 1)/(\gamma - 1)}$. Equation (7) can be solved for $M^*_{k,n}$ by several numerical methods. The method used in the program is Newton's method (ref. 8). The Mach number is then determined from

$$M_{k,n} = \sqrt{\frac{\left(\frac{2}{\gamma + 1}\right) M_{k,n}^{*2}}{1 - \left(\frac{\gamma - 1}{\gamma + 1}\right) M_{k,n}^{*2}}} \tag{8}$$

Thus, the Mach angle $\mu_{k,n}$ can be obtained from equation (6).

The slope of the characteristic lines of family I is $\tan(\mu + \varphi)$ and that of family II is $-\tan(\mu - \varphi)$ (ref. 7). However, since the regions are finite, a better estimate of these slopes is the slope based on the average value of the angles of the adjacent regions. Therefore,

$$m_I = \tan\left(\frac{\mu_{k,n} + \mu_{k-1,n+1}}{2} + \frac{\varphi_k + \varphi_{k-1}}{2}\right) \tag{9}$$

and

$$m_{II} = -\tan\left(\frac{\mu_{k,n} + \mu_{k+1,n}}{2} - \frac{\varphi_k + \varphi_{k+1}}{2}\right) \tag{10}$$

The x, y-coordinates of general intersection can now be determined. From analytical geometry,

$$x_{k,n} = \frac{\left(y_{k+1,n-1} - m_{II} x_{k+1,n-1}\right) - \left(y_{k-1,n} - m_I x_{k-1,n}\right)}{m_I - m_{II}} \tag{11}$$

and

$$y_{k,n} = y_{k-1,n} + m_I\left(x_{k,n} - x_{k-1,n}\right) \tag{12}$$

These equations are general and apply directly to the interior points. The special equations for the throat, nozzle contour, and centerline regions are given in the following section along with the procedure.

Calculation Procedure

The nozzle design is based on a throat half-height of 1. Therefore, the coordinates at the throat are $x = 0$ and $y = 1$. The region near the throat where $n = 1$ requires a special set of coordinate equations. At the point $n = 1$, $k = 1$, equations (11) and (12) reduce to

$$\left.\begin{aligned} x_{1,1} &= -\frac{1}{m_{II}} \\ y_{1,1} &= 0 \end{aligned}\right\} \qquad (13)$$

When $n = 1$ and $2 \leq k \leq (k_{max} - 1)$, equations (11) and (12) reduce to

$$x_{k,1} = \frac{1 - \left[y_{k-1,1} - m_I x_{k-1,1}\right]}{m_I - m_{II}} \qquad (14a)$$

$$y_{(k,1)} = y_{k-1,1} + m_I(x_{k,1} - x_{k-1,1}) \qquad (14b)$$

At the nozzle contour point where $n = 1$ and $k = k_{max}$, the contour must be shaped so that it cancels the waves. Therefore,

$$m_{II} = \tan\left(\varphi_{k_{max}, n}\right) \qquad (15)$$

Substituting equation (15) into equation (14a) and then solving equations (14a) and (14b) give the nozzle contour point for $n = 1$.

For all other values of n, $2 \leq n \leq (n_{max} - 1)$, the following coordinate equations are used. At the centerline of the nozzle where $k = 1$, $m_I = 0$. Therefore, equations (11) and (12) reduce to

$$x_{k,n} = -\frac{y_{k+1,n-1} - m_{II}\, x_{k+1,n-1}}{m_{II}} \qquad (16)$$

$$y_{k,n} = 0$$

For the interior points $2 \leq k \leq (k_{max} - 1)$, equations (11) and (12) are used. For the nozzle contour points $k = k_{max}$, equations (11) and (12) are used with equation (15) for m_{II}. As n increases by 1, k_{max} is reduced by 1 until $k_{max} = 2$. This procedure gives all the required nozzle contour points.

Increment Size

The increment in flow turning $\Delta\nu$ is the choice of the designer. The accuracy required determines the increment size used. The exit y-coordinate is known from one-dimensional isentropic flow and is equal to the area ratio A_e/A^*. Reference 5 suggests a method of extrapolation to obtain the exit x-coordinate for zero increment size. An example of this method is shown in figure 3, where the increment size $\Delta\nu$ is plotted against the exit x, y-coordinates. It is apparent from figure 3, that for an increment size of approximately 0.1^o, the errors in the exit x, y-coordinates are 0.5 percent and 0.02 percent, respectively. For a $\Delta\nu$ of approximately 0.5^o, the errors in the exit x, y-coordinates are 2.5 percent and 0.08 percent, respectively.



Figure 3. - An example effect of increment size on exit coordinates. Exit Mach number, 4.05; specific-heat ratio, 1.36.

DESCRIPTION OF INPUT

A description of the input for the FORTRAN IV computer program is given in this section. The input quantities are exit Mach number, increment in flow turning, specific-heat ratio, and an indicator controlling printout.

The input format with sample data is shown in table I.

TABLE I. - SAMPLE INPUT DATA

[Numbers in top row are card column numbers.]

1 6	7 12	13 18	19 20
ME	DV	GAM	NP
4.05	0.10	1.36	10

The input variables are

ME nozzle-exit Mach number, M_e

DV increment in flow turning, $\Delta\nu$, deg (this is an initial value; the program will adjust this value to make k_{max} an integer)

GAM specific-heat ratio, γ

NP nozzle coordinate printout indicator (for example, if NP = 1, every coordinate is printed out; if NP = 10, every tenth coordinate is printed out; it should also be noted that for any value of NP, the last coordinate is always printed out, and NP must be an integer, and right adjusted)

DESCRIPTION OF OUTPUT

A sample of the output obtained from the program is given in table II. The first line is input, except for VE, the Prandtl-Meyer angle based on the input exit Mach number, and DELTA V, the calculated increment in flow turning used in the program. The terms X and Y are the x, y-coordinates of the nozzle contour.

TABLE II. - SAMPLE-OUTPUT

TWO DIMENSIONAL SUPERSONIC NOZZLE WITH A SHARP-EDGED THROAT

VE = 69.929 ME = 4.050 GAMMA = 1.360 DELTA V = 0.09990

X	Y
0.	1.00000
1.92927	2.34001
2.40468	2.65508
2.82587	2.92363
3.24009	3.17764
3.66463	3.42788
4.10895	3.67942
4.57970	3.93519
5.08242	4.19708
5.62220	4.46641
6.20405	4.74416
6.83314	5.03110
7.51489	5.32781
8.25514	5.63474
9.06021	5.95222
9.93702	6.28043
10.89316	6.61945
11.93699	6.96920
13.07779	7.32944
14.32581	7.69973
15.69246	8.07943
17.19044	8.46761
18.83392	8.86304
20.63869	9.26413
22.62244	9.66883
24.80494	10.07457
27.20842	10.47817
29.85779	10.87570
32.78108	11.26233
36.00989	11.63222
39.57981	11.97823
43.53107	12.29177
47.90917	12.56245
52.76566	12.77779
58.15906	12.92276
64.15586	12.97935
64.15586	12.97935

PROGRAM DESCRIPTION

Main Program

All the calculations are made in the main program except the calculations for Mach number and Mach angle, which are made in Subroutine UA. The program variables for the main program are

I	$k + 2(n - 1)$
KMAX	integer (eq. (2)), k_{max}
NC	index for characteristic of family I, n
NK	number of Mach angles
NMAX	integer (eq. (3)), n_{max}
NMA	NMAX - 1
PHI(K)	flow angle, φ_k
SLOPE 1	slope of family II waves
SLOPE 2	slope of family I waves
U(I)	Mach angle, u_I
X(K, N)	x-coordinate
Y(K, N)	y-coordinate
XC(NC)	x-coordinate of contour point
YC(NC)	y-coordinate of contour point
XT	x-coordinate at throat
YT	y-coordinate at throat

Subroutine UA

This subroutine calculates the Mach number and Mach angle. Equation (7) is solved by Newton's method for $M^*_{k,n}$. The Mach number and Mach angle are then determined. The calling sequence for UA is as follows:

CALL UA(I, DV, GAM, U(I))

where

I	value of $k + 2(n - 1)$
DV	increment in flow turning (input), $\Delta\nu$

GAM specific-heat ratio (input), γ

U(I) Mach angle (output), μ_I

The program variables for UA are

DVDMC derivative of eq. (7), $d\nu/dM^*$

M Mach number, M

MC M^*

V Prandtl-Meyer angle, ν

VA(MC) function statement, where VA is eq. (7)

V1 temporary storage location for V

U Mach angle, μ

X2M limit on ν, ν_{max} - 0.01

```
$IBFTC NOZZLE  DECK

C      TWO DIMENSIONAL   SUPERSONIC NOZZLE WITH A SHARP-EDGED THROAT
       DIMENSION XC(1500),YC(1500),PHI(1500),U(3000),
      1X(1500,2),Y(1500,2)
       COMMON GAM
       REAL M,MC,ME,MEC
4      READ(5,3) ME,DV,GAM,NP
3      FORMAT(3F6.3,I2)
       VE=SQRT((GAM+1.0)/(GAM-1.0))*ATAN(SQRT((GAM-1.0)/(GAM+1.0)
      1*(ME**2-1.0))) - ATAN(SQRT(ME**2-1.0))
       DV= DV*.017453
       XT=0.0
       YT=1.0
       KMAX = INT(.5*VE/DV+1.5)
       DV=VE/(2.0*FLOAT(KMAX-1))
       NMAX = KMAX
       WRITE(6,251)
251    FORMAT(1H1,36X,59HTWO DIMENSIONAL SUPERSONIC NOZZLE WITH A SHARP-E
      1DGED THROAT)
       VE= VE*57.2958
       DV =DV*57.2958
       WRITE(6,252) VE,ME   ,GAM,DV
252    FORMAT(1H0,4HVE =,F8.3.10X,4HME =,F8.3,10X,7HGAMMA =,F6.3,10X,
      110HDELTA V = ,F7.5)
       WRITE(6,253)
253    FORMAT(1H0,5X,1HX,14X,1HY)
       WRITE (6,254) XT,YT
       DV= DV*.017453
C     FLOW ANGLE CALCULATION
       DO 5 K=1,KMAX
5      PHI(K) = FLOAT(K-1)*DV
C     MACH ANGLE CALCULATION
       NK = 2*KMAX-1
       DO 6 I=1,NK
6      CALL UA(I,DV,GAM,U(I))
C     REGION NEAR THE THROAT
       DO 100 K=1,KMAX
       NC = 1
       N=1
       I = K+2*(NC-1)
       IF (K.NE.1) GO TO 10
       SLOPE1 = -TAN((U(I)+U(I+1))/2.0-(PHI(K)+PHI(K+1))/2.0)
       X(K,N)= - 1.0/SLOPE1
       Y(K,N) = 0.0
       GO TO 100
10     IF (K.EQ.KMAX) GO TO 20
       SLOPE1 = -TAN((U(I)+U(I+1))/2.C-(PHI(K)+PHI(K+1))/2.0)
       SLOPE2 =  TAN((U(I)+U(I+1))/2.0+(PHI(K)+PHI(K-1))/2.0)
       X(K,N) = (1.0-(Y(K-1,N)-SLOPE2*X(K-1,N)))/
      1(SLOPE2-SLOPE1)
       Y(K,N)=Y(K-1,N)+SLOPE2*(X(K,N)-X(K-1,N))
       GO TO 100
```

```
20     SLOPE1= TAN(PHI(K))
       SLOPE2 =  TAN((U(I)+U(I+1))/2.0+(PHI(K)+PHI(K-1))/2.0)
       X(K,N) = (1.0-(Y(K-1,N)-SLOPE2*X(K-1,N)))/
      1(SLOPE2-SLOPE1)
       Y(K,N)=Y(K-1,N)+SLOPE2*(X(K,N)-X(K-1,N))
       XC(NC)=X(KMAX,N)
       YC(NC)= Y(KMAX,N)
100    CONTINUE
       IF (MOD(NC,NP) .EQ.0) WRITE(6,254) XC(NC),YC(NC)
C     REGION DOWNSTREAM OF THE THROAT
205    N=2
       KMAX=KMAX-1
       NC=NC+1
       DO 200  K=1,KMAX
       I = K+2*(NC-1)
       IF (K.NE.1) GO TO 201
       SLOPE1 = -TAN((U(I)+U(I+1))/2.0-(PHI(K)+PHI(K+1))/2.0)
       X(K,N)=-(Y(K+1,N-1)-SLOPE1*X(K+1,N-1))/SLOPE1
       Y(K,N)=0.0
       GO TO 200
201    IF(K.EQ.KMAX) GO TO 202
       SLOPE1 = -TAN((U(I)+U(I+1))/2.0-(PHI(K)+PHI(K+1))/2.0)
       SLOPE2 =  TAN((U(I)+U(I+1))/2.0+(PHI(K)+PHI(K-1))/2.0)
       X(K,N)=((Y(K+1,N-1)-SLOPE1*X(K+1,N-1))-(Y(K-1,N)-SLOPE2
      1*X(K-1,N)))/(SLOPE2 - SLOPE1)
       Y(K,N)=Y(K-1,N)+SLOPE2*(X(K,N)-X(K-1,N))
       GO TO 200
202    SLOPE1=TAN(PHI(K))
       SLOPE2 =  TAN((U(I)+U(I+1))/2.0+(PHI(K)+PHI(K-1))/2.0)
       X(K,N)=((Y(K+1,N-1)-SLOPE1*X(K+1,N-1))-(Y(K-1,N)-SLOPE2
      1*X(K-1,N)))/(SLOPE2 - SLOPE1)
       Y(K,N)=Y(K-1,N)+SLOPE2*(X(K,N)-X(K-1,N))
       XC(NC)=X(KMAX,N)
       YC(NC)=Y(KMAX,N)
200    CONTINUE
       IF (MOD(NC,NP) .EQ.0) WRITE(6,254) XC(NC),YC(NC)
       IF (KMAX.EQ.2) GO TO 250
       DO 204 K=1,KMAX
       X(K,1)=X(K,2)
204    Y(K,1)=Y(K,2)
       GO TO 205
250    NMA=NMAX-1
       WRITE (6,254) XC(NMA),YC(NMA)
254    FORMAT(1H ,F10.5,F15.5)
       GO TO 4
       END
```

```
$IBFTC U        DECK
      SUBROUTINE UA(I,DV,GAM,U)
      COMMON GAM
      REAL MC,M,MC1
      X2M=SQRT((GAM+1.0)/(GAM-1.0))-.01
      V = FLOAT(I-1)*DV
      IF(V.EQ.0.0) GO TO 5
      EXTERNAL  VA
      IF (MC.NE.1.0) GO TO 7
      DVDMC=.02
      GO TO 4
7     A= SQRT((GAM+1.0)/(GAM-1.0))
      B= (GAM-1.0)*MC**2-GAM
      C= (GAM+1.0)/MC**2-GAM
      DVDMC= A*(GAM-1.0)*MC/(1.0-B**2)**(.5)-MC**(-3)*(GAM+1.0)/(1.0-
     1C**2)**(.5)
4     MC= (V-VA(MC))/DVDMC+MC
      IF (MC.LT.X2M) GO TO 9
      MC=X2M
      WRITE (6,2) MC,NC,K
2     FORMAT(1H0,41HLIMIT HAS BEEN REACHED.   MC SET = TO X2M,5X,4HMC =,
     1F8.3,5X,4HNC =,I4,5X,3HK =,I4)
9     V1= VA(MC)
      IF (ABS(V-V1).LT. .00001) GO TO 8
      GO TO 7
8     M = SQRT(((2.0/(GAM+1.0))*MC**2)/(1.0-((GAM-1.0)/
     1(GAM+1.0)*MC**2)))
      GO TO 6
5     M=1.0
      MC=1.0
6     U=ARSIN(1.0/M)
      RETURN
      END
```

```
$IBFTC VAN      DECK
      FUNCTION VA(X)
      COMMON GAM
      A= SQRT((GAM+1.0)/(GAM-1.0))-1.0
      B= ARSIN((GAM-1.0)*X**2-GAM)
      C=ARSIN((GAM+1.0)/X**2-GAM)
      VA=(3.1415926)/4.0 * A+.5*((A+1.0)*B+C)
      RETURN
      END
```

Lewis Research Center,
National Aeronautics and Space Administration,
Cleveland, Ohio, October 18, 1967,
128-31-02-25-22.

REFERENCES

1. Vanco, Michael R.: Thermodynamic and Turbine Characteristics of Hydrogen-Fueled Open-Cycle Auxiliary Space Power Systems. NASA TM X-1337, 1967.
2. Goldman, Louis J.; and Scullin, Vincent, J.: Analytical Investigation of Supersonic Turbomachinery Blading. I - Computer Program for Blading Design. NASA Technical Note, estimated publication date, January, 1968.
3. Shapiro, A. H.; and Edelman, G. M.: Method of Characteristics for Two-Dimensional Supersonic Flow-Graphical and Numerical Procedures. Meteor Rep. No. 7, Massachusetts Inst. Tech., Feb. 1, 1947.
4. Moffitt, Thomas P.: Design and Experimental Investigation of a Single-Stage Turbine with a Rotor Intering Relative Mach Number of 2. NACA RM E58F202, 1958.
5. Edelman, Gilbert M.: The Design, Development, and Testing of Two-Dimensional Sharp-Cornered Supersonic Nozzles. Meteor Rep. No. 22, Massachusetts Inst. Tech., May 1, 1948.
6. Shames, Harold; and Seashore, Ferris L.: Design Data for Graphical Construction of Two-Dimensional Sharp-Edge-Throat Supersonic Nozzles. NACA RM E8J12, 1948.
7. Shapiro, Ascher H.: The Dynamics and Thermodynamics of Compressible Fluid Flow. Vol. I. Ronald Press Co., 1953.
8. Britton, Jack R.: Calculus. Holt, Rinehart and Winston, Inc., 1956.

NASA-Langley, 1968 — 1 E-4150

"The aeronautical and space activities of the United States shall be conducted so as to contribute . . . to the expansion of human knowledge of phenomena in the atmosphere and space. The Administration shall provide for the widest practicable and appropriate dissemination of information concerning its activities and the results thereof."

—NATIONAL AERONAUTICS AND SPACE ACT OF 1958

NASA SCIENTIFIC AND TECHNICAL PUBLICATIONS

TECHNICAL REPORTS: Scientific and technical information considered important, complete, and a lasting contribution to existing knowledge.

TECHNICAL NOTES: Information less broad in scope but nevertheless of importance as a contribution to existing knowledge.

TECHNICAL MEMORANDUMS: Information receiving limited distribution because of preliminary data, security classification, or other reasons.

CONTRACTOR REPORTS: Scientific and technical information generated under a NASA contract or grant and considered an important contribution to existing knowledge.

TECHNICAL TRANSLATIONS: Information published in a foreign language considered to merit NASA distribution in English.

SPECIAL PUBLICATIONS: Information derived from or of value to NASA activities. Publications include conference proceedings, monographs, data compilations, handbooks, sourcebooks, and special bibliographies.

TECHNOLOGY UTILIZATION PUBLICATIONS: Information on technology used by NASA that may be of particular interest in commercial and other non-aerospace applications. Publications include Tech Briefs, Technology Utilization Reports and Notes, and Technology Surveys.

Details on the availability of these publications may be obtained from:

SCIENTIFIC AND TECHNICAL INFORMATION DIVISION

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

Washington, D.C. 20546